Journal of Science Education Vol.25 n 2 2024

Journal of Science Education

ISSN 0124-5481

Modelling and Optimization of Brayton Cycle with Microsoft Excel

Pankaj Dumka[a*], Rishika Chauhan[b], Manoj Dubey[a], Dhananjay R. Mishra[a].

[a] Department of Mechanical Engineering, Jaypee University of Engineering and Technology, Guna, India

[b]Department of Electronics and Communication Engineering, Jaypee University of Engineering and Technology, Guna, India

ARTICLEINFO

Keywords:
Brayton Cycle,
Gas power cycle,
Microsoft Excel,
Thermodynamic analysis,
Spreadsheets, Excel system modelling

ABSTRACT

This article aims to demonstrate the effectiveness of utilizing the Microsoft Excel package for simulating the Brayton Cycle and its variations. Utilizing Microsoft Excel, the thermodynamic properties were derived verifying their value with existing literature. Energy balance calculations on individual components of the Brayton Cycle are performed to ascertain thermodynamic properties at various stages. The benefits of employing Microsoft Excel include its user-friendly interface, versatility in handling complex calculations, and compatibility with other software tools, thereby providing engineers with a powerful platform for comprehensive thermodynamic analysis and optimization.

1. Introduction

The Brayton Cycle stands as a fundamental concept in thermodynamics, serving as the operational principle behind gas turbines, jet engines, and other similar power generation systems. Its significance lies in its efficiency and widespread application across various industries. However, the complexity of the mathematical models often poses a challenge for students, engineers, and enthusiasts aiming to understand and apply Brayton Cycle principles in practical scenarios. Previous literature has extensively explored the Brayton Cycle and its applications in various contexts. Classical thermodynamics textbooks, such as those by Moran and Shapiro[z], Çengel and Boles[2] and P. K. Nag[3], provide in-depth theoretical explanations of the cycle along with mathematical derivations. While these resources offer valuable insights, they often require a strong mathematical background and may be daunting for those seeking a more practical, hands-on approach.

In recent years, there has been a growing trend towards utilizing computational tools for thermodynamic analysis. Software packages like Engineering Equation Solver (EES), MATLAB, and Python have been employed to simulate and analyse Brayton Cycle performance [4], [5], [6], [7], [8]. While these tools offer powerful capabilities, they may require specialized knowledge and licensing, limiting accessibility, particularly for students and enthusiasts.

Spreadsheets offer engineers a flexible and easy-to-use medium for crafting detailed models of intricate thermodynamic systems such as the Rankine Cycle. Widely used applications like Microsoft Excel and Google Sheets offer powerful computational capabilities, streamlining the process of simulating and analysing the impact of bleed pressure on critical performance metrics[9]. Numerous scholars have collaborated to create tools for tackling complex engineering challenges within spreadsheet environments.

El-hajj et al.[10] proposed a method for approximating numerical solutions to systems of nonlinear differential equations with a single variable, utilizing spreadsheets. Al-Awad [11]developed an Microsoft Excel Add-In for obtaining Refrigerants Properties and showcased its application in optimizing Multi-Stage Compression Refrigeration Cycles. Arganbright [12]introduced instructional techniques enabling educators to integrate animated graphical displays into spreadsheet constructions, thereby enriching mathematical understanding through engaging demonstrations. Musti and colleagues[13] demonstrated the creation of a Microsoft Excel-based Power System Load Flow Analysis tool for system planning and operation. Additionally, Musti[14] designed a Microsoft Excel-based tool for Power System Static State Estimation. El-Awad[15] introduced an Add-In for Microsoft Excel capable of determining thermodynamic properties for various fluids. Another contribution by El-Awad[16] involved developing a spreadsheet model using the effectiveness-NTU method. Fellah[17] utilized spreadsheets to model a plant based on exergy destruction, explicitly considering regenerator design factors such as size and overall heat-transfer coefficient. Sambaraju[18] presented a comprehensive methodology for modelling the Gauss-Newton Method for Non-Linear Data using Microsoft Excel. Razzak and Uddin[19] have used the spreadsheet to asses direct beam, diffused and global solar radiation. To bridge this gap, researchers have explored alternative platforms for thermodynamic modelling, including spreadsheet-based solutions. For instance, Maxner[20] performance of a combined

Accepted 1 October 2024, Available online 10 March 2025

cycle power plant. By utilizing built-in functions and macros, they simplified complex calculations while providing a user-friendly interface.

This paper proposes a solution to this challenge by introducing a user-friendly approach to Brayton Cycle modelling using Microsoft Excel. Leveraging the widespread accessibility and familiarity of Microsoft Excel, this method aims to simplify the process of understanding and implementing Brayton Cycle models. By harnessing the computational capabilities and visualization tools within Microsoft Excel, users can gain insights into the thermodynamic processes underlying the Brayton Cycle with ease. Therefore the novelty of the current study can be summarized as:

- The article demonstrates Microsoft Excel's effectiveness in simulating Brayton Cycles, showcasing its versatility for thermodynamic calculations. Leveraging Excel's Solver and Goal Seek, engineers optimize cycle parameters, offering a novel approach to comprehensive analysis.
- By utilizing Excel's Solver and Goal Seek, the study identifies optimal Brayton Cycle conditions, streamlining design processes. This highlights Excel's utility in engineering analysis and presents valuable contributions to thermodynamics.

2. Brayton Cycle layouts, Thermodynamics, and Microsoft Excel Modelling

In this section different types of Brayton Cycles will be explained. Then the different components are thermodynamically explained and finally the equations are solved in Microsoft Excel and plots are generated. In the analysis it is assumed that the specific heat at constant pressure for air is constant and the entropy at the compressor inlet is assumed to be 0.5 kJ/kg-K. The compressor inlet will always be the starting point for the cycle analysis.

2.1Simple Brayton Cycle

Fig. 1 shows the schematic of a simple Brayton Cycle. In ideal conditions the work interactions are isentropic, and the heat additions are isobaric. For the sake of simplicity, it has been assumed that in the pipelines connecting the components there is no frictional pressure drop. But, to introduce some complexity the turbine and compressors are only working adiabatically not isentropically. The input data will be the minimum temperatures, minimum pressure, maximum temperature, compression ratio, and turbine and compressor efficiencies.



Fig. 1: Schematic of simple Brayton Cycle

The theoretical T-s diagram of the cycle can be seen in the Fig. 2 based on which the thermodynamic analysis is done.



Fig. 2: Theoretical T-s diagram

i. Compressor (Process 1-2): First temperature at the exit of compression is evaluated considering isentropic conditions (2s). Then the enthalpy at 2s is evaluated then considering the compressor efficiency the final exit enthalpy is evaluated and based on it the exit temperature and entropies are obtained.

$$T_{2s} = T_1 \times r_p^{\frac{\gamma-1}{\gamma}} \tag{1}$$
$$h_{2s} = c_p \times T_{2s} \tag{2}$$
$$h_2 = h_1 + \frac{h_{2s}-h_1}{\eta_c} \tag{3}$$
$$T_2 = h_2/c_p \tag{4}$$
$$s_2 = s_{2s} + c_p \ln\left(\frac{T_2}{T_{2s}}\right) \tag{5}$$
$$W_c = h_2 - h_1 \tag{6}$$

ii. High temperature heat exchanger (Process 2-3): Here the heat is added isobarically. As the maximum cycle temperature is known so the task here is to calculate the exit enthalpy and entropy.

$$h_3 = c_p \times T_3 \tag{7}$$
$$s_3 = s_2 + c_p \ln\left(\frac{T_3}{T_2}\right) \tag{8}$$
$$Q_1 = h_3 - h_2 \tag{9}$$

iii. Turbine (Process 3-4): In the turbine as the expansion is not isentropic so first the isentropic exit temperature is obtained (T_{4s}) based on which the enthalpy h_{2s} is evaluated. Then considering the turbine efficiency the actual exit enthalpy (h_2) is evaluated. Based on which the actual exit temperature and entropies are obtained.

$$T_{4s} = T_3/r_p^{\frac{\gamma-1}{\gamma}} \tag{10}$$
$$h_{4s} = c_p \times T_{4s} \tag{11}$$
$$h_4 = h_3 - \eta_t \times (h_{2s} - h_1) \tag{12}$$
$$T_4 = h_4/c_p \tag{13}$$
$$s_4 = s_{4s} + c_p \ln\left(\frac{T_4}{T_{4s}}\right) \tag{14}$$
$$W_t = h_3 - h_4 \tag{15}$$

iv. Low temperature heat exchanger (Process 4-1):

$$Q_2 = h_4 - h_1 \tag{16}$$

Finally, the cycle efficiency can be evaluated as Eqn. 17.

$$\eta = (W_t - W_c)/Q_1 \tag{17}$$

To model this in Microsoft Excel consider the input data as shown in Tab. 1.

Tab. 1: Cycle parameters for simple Brayton Cycle

Parameter	Data	Unit
T_{min}	300	K
p_{min}	0.1	MPa
r_p	6.25	
T_{max}	1073	K
η_t	80	%
η_c	80	%
c_p	1.01	kJ/kgK
s_1(assumption)	0.5	kJkgK

γ	1.4

Now to model it in Microsoft Excel the input data section is created (Fig.3) and labelling of the data has been done.

Input Data		
T_min	300	K
p_min	0.1	Mpa
rp	6.25	
T_max	1073	K
eta_t	80	%
eta_c	80	%
cp	1.005	kJ/kgK
s_asmpt	0.5	
gamma	1.4	

Fig. 3: Input data

The calculations section is created based on the Eqn.'s 1-5,7-8, and 10-14. Fig. 4 shows the formulas in the sheet whereas, Fig. 4(b) displays the numerical values.

State	p	Ts	T	hs	h	s	ss
			Calculations				
1	=p_min		=T_min		=cp*I4	=s_asmpt	
2	=G4*rp	=I4*(rp)^((gamma-1)/gamma)	=K5/cp	=cp*H5	=K4+100*(J5-K4)/eta_c	=M5+cp*LN(I5/H5)	=L4
3	=G5		=T_max		=cp*I6	=L5+cp*LN(I6/I5)	
4	=G4	=I6/rp^((gamma-1)/gamma)	=K7/cp	=cp*H7	=K6-eta_t*(K6-J7)/100	=M7+cp*LN(I7/H7)	=L6

(a) Formulas in cells

State	p	Ts	T	hs	h	s	ss
			Calculations				
1	0.100		300.000		301.500	0.500	
2	0.625	506.425	558.032	508.958	560.822	0.598	0.500
3	0.625		1073.000		1078.365	1.255	
4	0.100	635.632	723.105	638.810	726.721	1.384	1.255

(b) Numerical values

Fig. 4: Calculations for different points

To ease up the calculations based on the Eqn.'s the enthalpies are labelled as shown in Fig. 5. Then based on Eqn.'s 6, 9, 15, 16, and 17 the energy interactions for different components along with the cycle efficiencies are evaluated as shown in Fig. 6.

h_1	=K4
h_2	=K5
h_3	=K6
h_4	=K7

Cell selection

h_1	301.500
h_2	560.822
h_3	1078.365
h_4	726.721

Numerical values

Fig. 5: Labelled enthalpies

Output	
w_t	=h_3-h_4
w_c	=h_2-h_1
Q_in	=h_3-h_2
Q_out	=h_4-h_1
w_net	=P3-P4
Q_et	=P5-P6
η	=100*P7/P5

(a) Formulas

Output		
w_t	351.644	kJ/kg
w_c	259.322	kJ/kg
Q_in	517.543	kJ/kg
Q_out	425.221	kJ/kg
w_net	92.322	kJ/kg
Q_et	92.322	kJ/kg
η	17.839	%

(b) Numerical Values

Fig. 6: Final calculations

Thereafter the T-s diagram is drawn using scatter plot and joining the points 1 to 4 one by one as shown in Fig. 7.



Fig. 7: T-s diagram for simple Brayton Cycle

Now comes the most important part i.e. finding the optimum pressure ratio for which the efficiency as well as work becomes maximum. Let us first do it manually i.e., varying the pressure ratio and evaluating the work and efficiency. The recorded data is shown in Tab. 2 (to show the plots in a clearer the turbine and compressor efficiencies are considered as 100%).

Tab. 2: Tabulation of net work output and efficiency for different pressure ratios

rp	w_net (kJ/kg)	η (%)	rp	w_net (kJ/kg)	η (%)
1	0.00	0.00	27	186.21	61.00
3	179.34	26.94	29	178.76	61.79
5	221.48	36.86	31	171.35	62.51
7	235.70	42.65	33	164.02	63.18
9	239.42	46.62	35	156.76	63.79
11	238.16	49.60	37	149.58	64.36
13	234.25	51.95	40	138.99	65.15
15	228.82	53.87	50	105.26	67.30
17	222.50	55.49	60	73.87	68.96
19	215.64	56.88	70	44.55	70.30
21	208.47	58.10	80	17.08	71.41
23	201.11	59.17	86	1.37	71.99

In Fig. 8, the work output of the cycle is plotted. It can be observed that the work output increases, reaching a maximum value somewhere within a compression ratio range of 9.1 to 9.5, after which it decreases. Meanwhile, the efficiency shows a continuous increase, albeit at a diminishing rate.



Fig. 8: Variation of net work output and efficiency of cycle as a function of pressure ratio

This phenomenon occurs because the cycle area increases with the rise in pressure ratio, reaching a maximum value before decreasing and eventually becoming zero. Additionally, as the cycle efficiency increases, it reaches a maximum value close to the Carnot efficiency. However, it's important to note that at this maximum efficiency point, the work output is zero. These observations are further supported by the T-s plots depicted in Fig. 9.



Fig. 9: T-s diagram at different pressure ratios

To optimize the cycle using Microsoft Excel's built-in functions, we'll utilize the powerful Solver function for single-objective optimization. One can find Solver under the Data tab in Excel. Once in the Solver dialogue box, specify the target cell as the net work output (to be maximized) and vary the pressure ratio (rp). We'll use the GRG nonlinear scheme for optimization, which can handle nonlinear problems efficiently. The dialogue box settings are illustrated in Fig. 10. For better readability, the cell containing the net work output data has already been labelled.



Fig. 10: Dialogue box for solver with variables

The Solver output reveals that the optimum value of the pressure ratio (rp) is 9.302, resulting in a maximum work output of 239.467. This aligns precisely with our manual calculations, as depicted in Fig. 11.



Input Data

T_min	300	K
p_min	0.1	Mpa
rp	9.302	
T_max	1073	K
eta_t	100	%
eta_c	100	%
cp	1.005	kJ/kgK
s_asmpt	0.5	
gamma	1.4	

Calculations

State	p	Ts	T	hs	h	s	ss
1	0.100		300.000		301.500	0.500	
2	0.930	567.358	567.358	570.195	570.195	0.500	0.500
3	0.930		1073.000		1078.365	1.140	
4	0.100	567.366	567.366	570.203	570.203	1.140	1.140

Output

w_t	508.162	kJ/kg
w_c	268.695	kJ/kg
Q_in	508.170	kJ/kg
Q_out	268.703	kJ/kg
w_net	239.467	kJ/kg
Q_et	239.467	kJ/kg
η	47.123	%

h_1	301.500
h_2	570.195
h_3	1078.365
h_4	570.203

Fig. 11: Cycle parameters for optimized Brayton Cycle using Solver

2.2 Brayton Cycle with reheating:

In case of reheating after expansion in the first turbine the air moves to the second heat exchanger for heating after which it again expands. The net work output will increase due to the reheating. The schematic diagram of the Brayton Cycle with reheat is shown in Fig. 12 (a) and the theoretical T-s diagram for the cycle is shown in Fig. 12 (b).



(a) Schematic diagram



(b) T-S diagram

Fig. 12: Schematic and theoretical T-s diagram for reheat Brayton Cycle

The thermodynamic analysis for all the components will remain the same as for simple Brayton Cycle with a difference that the heat added, and turbine will be different now as expressed in Eqn.'s 18-19.

$$W_t = W_{t1} + W_{t2} = (h_3 - h_4) + (h_5 - h_6) \tag{18}$$

$$Q_{added} = Q_1 + Q_1^{'} \tag{19}$$

To model this cycle in Microsoft Excel, consider the input data as shown in Tab. 2.

Tab. 2: Cycle parameters for simple reheated Brayton Cycle

Parameter	Data	Unit
T_{min}	300	K
p_{min}	0.1	Mpa
$r_{p,c}$	9	
$r_{p,t(\text{for both})}$	3	
T_{max}	1000	K
η_t	100	%
η_c	100	%
c_p	1.01	kJ/kgK
s_1(assumption)	0.5	kJ/kgK
γ	1.4	

The input data section is now established in Microsoft Excel, as shown in Fig. 13, and the data has been labelled accordingly.

	B	C	D
2	Input Data		
3	T_min	300	K
4	p_min	0.1	Mpa
5	rp	9	
6	rp_t	3	
7	T_max	1000	K
8	eta_t	100	%
9	eta_c	100	%
10	cp	1.005	kJ/kgK
11	s_asmpt	0.5	
12	gamma	1.4	

Fig. 13: Input data

The calculations section has been constructed using the same equations as simple Brayton Cycle case. Fig. 14(a) illustrates the formulas utilized in the sheet, while Fig. 14 (b) exhibits the corresponding numerical values.

	F	G	H	I	J	K	L	M
2				Calculations				
3	State	p	Ts	T	hs	h	s	ss
4	1	=p_min		=T_min		=cp*I4	=s_asmpt	
5	2	=G4*rp	=I4*(rp)^((gamma-1)/gamma)	=K5/cp	=cp*H5	=K4+100*(J5-K4)/eta_c	=M5+cp*LN(I5/H5)	=L4
6	3	=G5		=T_max		=cp*I6	=L5+cp*LN(I6/I5)	
7	4	=G6/rp_t	=I6/rp_t^((gamma-1)/gamma)	=K7/cp	=cp*H7	=K6-eta_t*(K6-J7)/100	=M7+cp*LN(I7/H7)	=L6
8	5	=G7		=T_max		=cp*I8	=L7+cp*LN(I8/I7)	
9	6	=p_min	=I8/rp_t^((gamma-1)/gamma)	=K9/cp	=cp*H9	=K8-eta_t*(K8-J9)/100	=M9+cp*LN(I9/H9)	=L8

(a) Formulas in cells

	F	G	H	I	J	K	L	M
2				Calculations				
3	tat	p	Ts	T	hs	h	s	ss
4	1	0.100		300.000		301.500	0.500	
5	2	0.900	562.033	562.033	564.843	564.843	0.500	0.500
6	3	0.900		1000.000		1005.000	1.079	
7	4	0.300	730.600	730.600	734.253	734.253	1.079	1.079
8	5	0.300		1000.000		1005.000	1.395	
9	6	0.100	730.600	730.600	734.253	734.253	1.395	1.395

(b) Numerical values

Fig. 14: Calculations for different points

To simplify the calculations according to the equations, the enthalpies are labelled as depicted in Fig. 15. Subsequently, the energy interactions for various components are assessed, along with the determination of cycle efficiencies, as demonstrated in Fig. 16.

	N	O
12	h_1	=K4
13	h_2	=K5
14	h_3	=K6
15	h_4	=K7
16	h_5	=K8
17	h_6	=K9

Cell selection

	N	O
12	h_1	301.500
13	h_2	564.843
14	h_3	1005.000
15	h_4	734.253
16	h_5	1005.000
17	h_6	734.253

Numerical values

Fig. 15: Labelled enthalpies

	O	P
2	Output	
3	w_t	=h_3-h_4+h_5 -h_6
4	w_c	=h_2-h_1
5	Q_in	=h_3-h_2+h_5-h_4
6	Q_out	=h_4-h_1
7	w_net	=P3-P4
8	Q_et	=P5-P6
9	η	=100*P7/P5

(c) Formulas

	O	P	Q
2	Output		
3	w_t	541.494	kJ/kg
4	w_c	263.343	kJ/kg
5	Q_in	710.904	kJ/kg
6	Q_out	432.753	kJ/kg
7	w_net	278.151	kJ/kg
8	Q_et	278.151	kJ/kg
9	η	39.126	%

(d) Numerical Values

Fig. 16: Final calculations

Following that, the T-s diagram is created by employing a scatter plot and connecting points 1 through 6 sequentially, as illustrated in Fig. 17.



Fig. 17: T-s diagram for reheat Brayton Cycle

If one want, then the problem can be extended further to understand the behaviour of reheat pressure on the cycle efficiency and net work output transfer.

2.3 Brayton Cycle with regeneration

Fig. 18 shows the schematic of Brayton Cycle with regeneration. Regeneration basically means the utilization of energy which is going to be exhausted in the heat exchanger for heating the compressed air before it enters the primary heat exchanger (the one where heat addition to the cycle takes place). Fig. 19 shows the theoretical T-s diagram for this cycle.



Fig. 18: Schematic diagram of regenerative Brayton Cycle



Fig. 19: Theoretical T-s diagram for regenerative Brayton Cycle

In the case of a regenerator in the Brayton cycle, a noticeable alteration occurs where heat addition begins at point 3 instead of point 2, and heat rejection starts at point 6 rather than point 5. Consequently, due to regeneration, a lower amount of heat addition is required for the same work transfer, resulting in an increase in cycle efficiency. An essential parameter to consider with a regenerator is its effectiveness, which quantifies the ratio of actual heat gain by the air to the maximum possible heat gain. This effectiveness is mathematically expressed in Eqn. 20.

$$\epsilon = (T_3 - T_2)/(T_5 - T_2) \quad (20)$$

To model this cycle in Microsoft Excel, consider the input data as shown in Tab. 3.

Tab. 3: Cycle parameters for simple regenerative Brayton Cycle

Parameter	Value	Unit
T_{min}	283	K
p_{min}	0.1013	MPa
r_p	5.5	
T_{max}	1023	K
η_t	85	%
η_c	82	%
c_p	1.005	kJ/kgK
s_1 (assumed)	0.5	kJ/kgK
γ	1.4	
ϵ	70	%

The Microsoft Excel file now includes a designated section for input data, as depicted in Fig. 20, with appropriate labels assigned to the data.

	B	C	D	E
2	Input Data			
3	T_min	283	K	
4	p_min	0.101	Mpa	
5	rp	5.5		
6	T_max	1023	K	
7	eta_t	85	%	
8	eta_c	82	%	
9	cp	1.005	kJ/kgK	
10	s_asmpt	0.5		
11	gamma	1.4		
12	epsilon	70	%	

Fig. 20: Input data

The calculations section has been set up with identical equations to those used in the simple Brayton Cycle case except point 3. Here the temperature T_3 is evaluated based on effectiveness of the regenerator (Eqn. 20). Fig. 21(a) displays the formulas employed in the sheet, while Fig. 21 (b) presents the associated numerical values.

	F	G	H	I	J	K	L	M
2				Calculations				
3	State	p	Ts	T	hs	h	s	ss
4	1	=p_min		=T_min		=cp*I4	=s_asmpt	
5	2	=G4*rp	=I4*(rp)^((gamma-1)/gamma)	=K5/cp	=cp*H5	=K4+100*(J5-K4)/eta_c	=M5+cp*LN(I5/H5)	=L4
6	3	=G5		=I5+epsilon*(I8-I5)/100		=cp*I6	=L5+cp*LN(I6/I5)	
7	4	=G5		=T_max		=cp*I7	=L5+cp*LN(I7/I5)	
8	5	=G4	=I7/rp^((gamma-1)/gamma)	=K8/cp	=cp*H8	=K7-eta_t*(K7-J8)/100	=M8+cp*LN(I8/H8)	=L7
9	6	=p_min		=K9/cp		=K4+P6	=L4+cp*LN(I9/I4)	

(a) Formulas in cells

	F	G	H	I	J	K	L	M
2				Calculations				
3	State	p	Ts	T	hs	h	s	ss
4	1	0.101		283.000		284.415	0.500	
5	2	0.557	460.594	499.579	462.897	502.076	0.582	0.500
6	3	0.557		631.279		634.435	0.817	
7	4	0.557		1023.000		1028.115	1.302	
8	5	0.101	628.555	687.722	631.698	691.161	1.392	1.302
9	6	0.101		554.663		557.437	1.176	

(b) Numerical values

Fig. 21: Calculations for different points

For streamlining the calculations based on the equations, the enthalpies are designated as illustrated in Fig. 22. Following this, the energy exchanges for different components are evaluated, along with the determination of cycle efficiencies, as shown in Fig. 23.

	N	O
14	h_1	=K4
15	h_2	=K5
16	h_3	=K6
17	h_4	=K7
18	h_5	=K8
19	h_6	=K9

Cell selection

	N	O
14	h_1	284.415
15	h_2	502.076
16	h_3	634.435
17	h_4	1028.115
18	h_5	691.161
19	h_6	558.802

Numerical values

Fig. 22: Labelled enthalpies

Output	
w_t	=h_4-h_5
w_c	=h_2-h_1
Q_in	=h_4-h_3
Q_out	=-P7+P5
w_net	=P3-P4
Q_et	=P5-P6
η	=100*P7/P5

(a) Formulas

Output		
w_t	336.954	kJ/kg
w_c	217.661	kJ/kg
Q_in	393.680	kJ/kg
Q_out	274.387	kJ/kg
w_net	119.293	kJ/kg
Q_et	119.293	kJ/kg
η	30.302	%

(b) Numerical Values

Fig. 23: Final calculations

Afterwards, the T-s diagram is generated by using a scatter plot and connecting points 1 through 6 sequentially, as demonstrated in Fig. 24.



Fig. 24: T-s diagram for regenerative Brayton Cycle

On fitting the cycle efficiency vs, the regenerator effectiveness, the following trend is observed as shown in Fig. 25.



Fig. 25: Variation of Brayton Cycle efficiency with the regenerator effectiveness

2.4 Brayton Cycle with intercooled compression, regeneration, and reheat

Intercooled compression is normally done to reduce the work input of compressor but to gain from it regeneration and reheating has to be done. Fig. 26 shows the schematic of such Brayton Cycle. This cycle is the most general Brayton Cycle.



Fig. 26: Schematic diagram of Brayton Cycle with regenerator and intercooler

The input data for the cycle is shown in Table 4 whereas Fig. 27 shows the same in the input section of the Microsoft Excel file.

Tab. 4: Cycle parameters for Generalized Brayton Cycle

Parameter	Value	Unit
T_{min}	300	K
p_{min}	0.1	MPa
$r_{p,c}$	3	
$r_{p,t}$	3	
T_{max}	1000	K
η_t	80	
η_c	80	
ϵ	80	
c_p	1.01	kJ/kg-K
s_1 (assumed)	0.5	kJ/kg-K
γ	1.4	

Input Data		
T_min	300	K
p_min	0.1	mPa
rp_c	3	
rp_t	3	
T_max	1000	K
eta_t	80	
eta_c	80	
epsilon	80	
cp	1.01	kJ/kg-K
s_asmpt	0.5	kJ/kg-K
gamma	1.4	

Fig. 27: Input data

The Microsoft Excel formulas are shown in Fig. 28 and the numerical values and program are shown in Fig. 29.



Calculations							
State	p	Ts	T	hs	h	s	ss
1	=p_min		=T_min		=cp*J3	=s_asmpt	
2	=H3*rp_c	=J3*rp_c^((gamma-1)/gamma)	=L4/cp	=cp*I4	=L3+100*(K4-L3)/eta_c	=N4+cp*LN(J4/I4)	=M3
3	=H4		=J3		=cp*J5	=M4+cp*LN(J5/J4)	
4	=H5*rp_c	=J5*rp_c^((gamma-1)/gamma)	=L6/cp	=cp*I6	=L5+100*(K6-L5)/eta_c	=N6+cp*LN(J6/I6)	=M5
5	=H6		=J6+epsilon*(J11-J6)/100		=cp*J7	=M6+cp*LN(J7/J6)	
6	=H6		=T_max		=cp*J8	=M6+cp*LN(J8/J6)	
7	=H8/rp_t	=J8/rp_t^((gamma-1)/gamma)	=L9/cp	=cp*I9	=L8-eta_t*(L8-K9)/100	=N9+cp*LN(J9/I9)	=M8
8	=H9		=T_max		=cp*J10	=M9+cp*LN(J10/J9)	
9	=H3	=J10/rp_t^((gamma-1)/gamma)	=L11/cp	=cp*I11	=L10-eta_t*(L10-K11)/100	=N11+cp*LN(J11/I11)	=M10
10	=H11		=L12/cp		=Q5-(h_2-h_3)+h_1	=M11+cp*LN(J12/J11)	

Output	
w_t	=h_6-h_7+h_8-h_9
w_c	=h_2-h_1+h_4-h_3
Q_in	=h_6-h_5+h_8-h_7
Q_out	=Q4-Q6
w_net	=Q2-Q3
Q_net	=Q6
η	=100*Q6/Q4

h_1	=L3
h_2	=L4
h_3	=L5
h_4	=L6
h_5	=L7
h_6	=L8
h_7	=L9
h_8	=L10
h_9	=L11
h_10	=L12

Fig. 28: Formulas in Microsoft Excel sheet



Calculations							
State	p	Ts	T	hs	h	s	ss
1.000	0.100		300.000		301.500	0.500	
2.000	0.300	410.621	438.277	412.675	440.468	0.566	0.500
3.000	0.300		300.000		301.500	0.185	
4.000	0.900	410.621	438.277	412.675	440.468	0.250	0.185
5.000	0.900		715.239		718.816	0.742	
6.000	0.900		1000.000		1005.000	1.079	
7.000	0.300	730.600	784.480	734.253	788.402	1.151	1.079
8.000	0.300		1000.000		1005.000	1.395	
9.000	0.100	730.600	784.480	734.253	788.402	1.466	1.395
10.000	0.100		507.517		510.055	1.028	

Output		
w_t	433.195	kJ/kg
w_c	277.936	kJ/kg
Q_in	502.782	kJ/kg
Q_out	347.523	kJ/kg
w_net	155.259	kJ/kg
Q_net	155.259	kJ/kg
η	30.880	%

h_1	301.500
h_2	440.468
h_3	301.500
h_4	440.468
h_5	718.816
h_6	1005.000
h_7	788.402
h_8	1005.000
h_9	788.402
h_10	510.055

Fig. 29: Program output and numerical magnitudes of parameters

The T-s diagram for this generalized Brayton Cycle is shown in Fig. 30.



Fig. 30: T-s diagram for the generalized Brayton Cycle

References

[1] M. J. Moran, H. N. Shapiro, D. D. Boettner, and M. B. Bailey, Fundamentals of engineering thermodynamics. John Wiley \& Sons, 2010.
[2] Y. A. Cengel, M. A. Boles, and M. Kano\uglu, Thermodynamics: an engineering approach, vol. 5. McGraw-hill New York, 2011.
[3] P. K. Nag, Engineering thermodynamics. Tata McGraw Hill, 2013.
[4] P. Dumka, R. Chauhan, A. Singh, G. Singh, and D. Mishra, "Implementation of Buckingham ' s Pi theorem using Python," Adv. Eng. Softw., vol. 173, no. July, p. 103232, 2022, doi: 10.1016/j.advengsoft.2022.103232.
[5] P. Dumka, N. Samaiya, S. Gandhi, and D. R. Mishra, "Modelling of Hardy Cross Method for Pipe Networks," Int. J. Mech. Eng., vol. 10, no. 2, pp. 1–8, 2023.
[6] K. Gajula, V. Sharma, B. Sharma, D. R. Mishra, and P. Dumka, "Modelling of Energy in Transit Using Python," Int. J. Innov. Sci. Res. Technol., vol. 7, no. 8, pp. 1152–1156, 2022.
[7] A. R. Joshi, A. Deo, A. Parashar, D. R. Mishra, and P. Dumka, "Modelling Steam Power Cycle using Python," Int. J. Sci. Res. Comput. Sci. Eng. Inf. Technol., vol. 9, no. 1, pp. 152–162, 2023.
[8] A. Darnal, M. Fisseler, C. C. Benjamin, and A. R. Srinivasa, "An Excel-based approach for designing stepped driveshafts for mobility devices and a study of its use in a design of machine elements course," Comput. Appl. Eng. Educ., no. August 2021, pp. 1–17, 2022, doi: 10.1002/cae.22528.
[9] J. M. Lindquist and C. A. Sulewski, "Microsoft Excel: The Universal Tool of Analysis," in Handbook of Military and Defense Operations Research, Chapman and Hall/CRC, 2020, pp. 19–54.
[10] A. El-Hajj, "Functional Simulation Using Spreadsheets," Simulation, vol. 73, pp. 80–90, 1999, [Online]. Available: https://api.semanticscholar.org/CorpusID:38215053
[11] M. M. El-Awad, "An Excel Add-In for Refrigerants Properties and its Use for Optimising Multi-Stage Compression Refrigeration Cycles," vol. 12, no. 1, pp. 1–18, 2003, [Online]. Available: https://sie.scholasticahq.com
[12] D. Arganbright, "Enhancing Mathematical Graphical Displays in Excel through Animation," Spreadsheets Educ., vol. 2, no. 1, pp. 1–25, 2006.
[13] M. K. S. Sastry, R. Roopchand, and R. Chhetri, "Power distribution system load flow using microsoft excel," Proc. - 2014 4th Int. Conf. Commun. Syst. Netw. Technol. CSNT 2014, vol. 6, no. 1, pp. 954–959, 2014, doi: 10.1109/CSNT.2014.196.
[14] M. K. S. Sastry, B. P.V., and R. R. Malireddy, "Power System State Estimation using Microsoft Excel," Spreadsheets Educ., vol. 8, p. 1, 2014, [Online]. Available: https://api.semanticscholar.org/CorpusID:53553993
[15] M. M. El-Awad, "A Multi-Substance Add-in for the Analyses of Thermo-fluid Systems using Microsoft Excel," Int. J. Eng. Appl. Sci., vol. 2, no. 3, 2015.
[16] M. Al-Awad, "Modelling the Regenerator for Multi-Objective Optimisation of the Air-Bottoming Cycle with Excel," Spreadsheets Educ., vol. 9, no. 2, pp. 1–13, 2016.
[17] G. Fellah, "Excel Spreadsheet as a Tool for Simulating the Performance of Steam Power Plants," vol. 12, no. January, pp. 1–19, 2019.
[18] P. Sambaraju, "Gauss-Newton Method for Non-Linear Data Using Microsoft Excel.," Spreadsheets Educ., pp. 1–9, 2023, doi: https://doi.org/10.1000/189.
[19] A. Razzak, Z. Uddin, and M. J. Iqbal, "New mathematical approach to determine solar radiation for the southwestern coastline of Pakistan," Adv. Energy Res., vol. 8, no. 2, pp. 111–123, 2022.
[20] M. R. Maixner, "ExcelTM analysis of combined cycle power plant," ASEE Annu. Conf. Expo. Conf. Proc., pp. 6077–6083, 2005, doi: 10.18260/1-2--14287.